// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: contracts/interfaces/IKannaStockOption.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.17;

import {IERC165} from "@openzeppelin/contracts/utils/introspection/IERC165.sol";

/**
 *   __
 *  |  | ___\|/_     ____    ____  _\|/_
 *  |  |/ /\__  \   /    \  /    \ \__  \
 *  |    <  / __ \_|   |  \|   |  \ / __ \_
 *  |__|_ \(____  /|___|  /|___|  /(____  /
 *       \/     \/      \/      \/      \/
 *            __                    __                       __   .__
 *    _______/  |_   ____    ____  |  | __   ____  ______  _/  |_ |__|  ____    ____
 *   /  ___/\   __\ /  _ \ _/ ___\ |  |/ /  /  _ \ \____ \ \   __\|  | /  _ \  /    \
 *   \___ \  |  |  (  <_> )\  \___ |    <  (  <_> )|  |_> > |  |  |  |(  <_> )|   |  \
 *  /____  > |__|   \____/  \___  >|__|_ \  \____/ |   __/  |__|  |__| \____/ |___|  /
 *       \/                     \/      \/         |__|                            \/
 *
 *  @title KNN Stock Option (Vesting)
 *  @author KANNA Team
 *  @custom:github  https://github.com/kanna-coin
 *  @custom:site https://kannacoin.io
 *  @custom:discord https://discord.kannacoin.io
 */
interface IKannaStockOption is IERC165 {
    enum Status {
        Cliff,
        Lock,
        Vesting
    }

    function totalVested() external view returns (uint256);
    function vestingForecast(uint256 date) external view returns (uint256);
    function availableToWithdraw() external view returns (uint256);
    function status() external view returns (Status);
    function withdraw(uint256 amountToWithdraw) external;
    function finalize() external;
    function abort() external;
}

// File: contracts/KannaStockOption.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.21;

import {IERC20} from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import {ReentrancyGuard} from "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import {IKannaStockOption} from "./interfaces/IKannaStockOption.sol";

/**
 *   __
 *  |  | ___\|/_     ____    ____  _\|/_
 *  |  |/ /\__  \   /    \  /    \ \__  \
 *  |    <  / __ \_|   |  \|   |  \ / __ \_
 *  |__|_ \(____  /|___|  /|___|  /(____  /
 *       \/     \/      \/      \/      \/
 *            __                    __                       __   .__
 *    _______/  |_   ____    ____  |  | __   ____  ______  _/  |_ |__|  ____    ____
 *   /  ___/\   __\ /  _ \ _/ ___\ |  |/ /  /  _ \ \____ \ \   __\|  | /  _ \  /    \
 *   \___ \  |  |  (  <_> )\  \___ |    <  (  <_> )|  |_> > |  |  |  |(  <_> )|   |  \
 *  /____  > |__|   \____/  \___  >|__|_ \  \____/ |   __/  |__|  |__| \____/ |___|  /
 *       \/                     \/      \/         |__|                            \/
 *
 *  @title KNN Stock Option (Vesting)
 *  @author KANNA Team
 *  @custom:github  https://github.com/kanna-coin
 *  @custom:site https://kannacoin.io
 *  @custom:discord https://discord.kannacoin.io
 */
contract KannaStockOption is IKannaStockOption, Ownable, ReentrancyGuard {
    IERC20 _token;
    uint256 _startDate;
    uint256 _daysOfVesting;
    uint256 _daysOfCliff;
    uint256 _daysOfLock;
    uint256 _percentOfGrant;
    uint256 _amount;
    address _beneficiary;
    uint256 _grantAmount;
    uint256 _withdrawn;
    uint256 _lockEndDate;
    uint256 _cliffEndDate;
    uint256 _vestingEndDate;
    bool _finalized;
    uint256 _finalizedAt;
    bool _initialized;
    uint256 _initializedAt;
    uint256 _lastWithdrawalTime;

    event Initialize(
        address tokenAddress,
        uint256 startDate,
        uint256 daysOfVesting,
        uint256 daysOfCliff,
        uint256 daysOfLock,
        uint256 percentOfGrant,
        uint256 amount,
        address beneficiary,
        uint256 initializedAt
    );

    event Withdraw(address indexed beneficiary, uint256 amount, uint256 elapsed);

    event Finalize(address indexed initiator, uint256 amount, uint256 elapsed);

    event Abort(address indexed beneficiary, uint256 amount);

    function initialize(
        address tokenAddress,
        uint256 startDate,
        uint256 daysOfVesting,
        uint256 daysOfCliff,
        uint256 daysOfLock,
        uint256 percentOfGrant,
        uint256 amount,
        address beneficiary
    ) external {
        if (owner() != address(0)) {
            _checkOwner();
        }
        require(_initialized == false, "KannaStockOption: contract already initialized");
        require(startDate > 0, "KannaStockOption: startDate is zero");
        require(daysOfVesting > 0, "KannaStockOption: daysOfVesting is zero");
        require(amount > 0, "KannaStockOption: amount is zero");
        require(beneficiary != address(0), "KannaStockOption: beneficiary is zero");
        require(
            daysOfCliff + daysOfLock <= daysOfVesting,
            "KannaStockOption: daysOfCliff plus daysOfLock overflows daysOfVesting"
        );
        require(percentOfGrant <= 100, "KannaStockOption: percentOfGrant is greater than 100");

        _token = IERC20(tokenAddress);

        require(_token.allowance(_msgSender(), address(this)) >= amount, "KannaStockOption: insufficient allowance");
        require(_token.transferFrom(_msgSender(), address(this), amount), "KannaStockOption: insufficient balance");

        _startDate = startDate;
        _daysOfVesting = daysOfVesting;
        _daysOfCliff = daysOfCliff;
        _daysOfLock = daysOfLock;
        _percentOfGrant = percentOfGrant;
        _amount = amount;
        _beneficiary = beneficiary;

        _grantAmount = (_amount * _percentOfGrant) / 100;

        _cliffEndDate = startDate + (daysOfCliff * 1 days);
        _lockEndDate = _cliffEndDate + (daysOfLock * 1 days);
        _vestingEndDate = startDate + (daysOfVesting * 1 days);

        _finalized = false;
        _withdrawn = 0;
        _initialized = true;
        _initializedAt = block.timestamp;

        emit Initialize(
            tokenAddress,
            startDate,
            daysOfVesting,
            daysOfCliff,
            daysOfLock,
            percentOfGrant,
            amount,
            beneficiary,
            block.timestamp
        );

        if (owner() == address(0)) {
            _transferOwnership(_msgSender());
        }
    }

    function timestamp() public view returns (uint256) {
        return _finalized ? _finalizedAt : block.timestamp;
    }

    function totalVested() public view initialized returns (uint256) {
        if (timestamp() < _cliffEndDate) return 0;
        if (timestamp() >= _vestingEndDate) return _amount;

        return (_amount * (timestamp() - _startDate)) / (_vestingEndDate - _startDate);
    }

    function vestingForecast(uint256 date) public view initialized returns (uint256) {
        require(date >= _startDate, "KannaStockOption: date is before startDate");

        if (date < _cliffEndDate) return 0;
        if (date >= _vestingEndDate) return _amount;

        return (_amount * (date - _startDate)) / (_vestingEndDate - _startDate);
    }

    function availableToWithdraw() public view initialized returns (uint256) {
        if (timestamp() < _cliffEndDate) return 0;
        if (timestamp() >= _vestingEndDate) return _amount - _withdrawn;

        if (block.timestamp < _lockEndDate && totalVested() > _grantAmount) return _grantAmount - _withdrawn;

        return totalVested() - _withdrawn;
    }

    function finalize() public nonReentrant initialized {
        require(
            _msgSender() == owner() || _msgSender() == _beneficiary,
            "KannaStockOption: caller is not the owner or beneficiary"
        );
        require(_finalized == false, "KannaStockOption: contract already finalized");

        uint256 availableAmount = availableToWithdraw();

        if (availableAmount > 0) {
            _withdrawn += availableAmount;
            _token.transfer(_beneficiary, availableAmount);
        }

        uint256 leftover = _amount - totalVested();

        if (leftover > 0) {
            _token.transfer(owner(), leftover);
        }

        _finalizedAt = block.timestamp;
        _finalized = true;

        emit Finalize(_msgSender(), availableAmount, _finalizedAt - _initializedAt);
    }

    function status() public view returns (Status) {
        if (timestamp() < _cliffEndDate) return Status.Cliff;
        if (timestamp() >= _vestingEndDate) return Status.Vesting;

        return Status.Lock;
    }

    function maxGrantAmount() public view returns (uint256) {
        return _grantAmount;
    }

    function withdraw(uint256 amountToWithdraw) public nonReentrant initialized {
        require(_msgSender() == _beneficiary, "KannaStockOption: caller is not the beneficiary");
        require(amountToWithdraw > 0, "KannaStockOption: invalid amountToWithdraw");
        require(
            amountToWithdraw <= availableToWithdraw(),
            "KannaStockOption: amountToWithdraw is greater than availableToWithdraw"
        );

        uint256 withdrawDate = block.timestamp;

        require(withdrawDate - _lastWithdrawalTime >= 1 days, "KannaStockOption: Only one withdrawal allowed per day");

        _withdrawn += amountToWithdraw;
        _token.transfer(_beneficiary, amountToWithdraw);

        emit Withdraw(_msgSender(), amountToWithdraw, withdrawDate - _initializedAt);

        if (_withdrawn == _amount) {
            _finalizedAt = withdrawDate;
            _finalized = true;

            emit Finalize(_msgSender(), amountToWithdraw, _finalizedAt - _initializedAt);
        }

        _lastWithdrawalTime = withdrawDate;
    }

    function abort() public nonReentrant {
        require(_token.balanceOf(address(this)) > 0, "KannaStockOption: contract has no balance");
        require(_msgSender() == _beneficiary, "KannaStockOption: caller is not the beneficiary");

        uint256 returnedAmount = _token.balanceOf(address(this));

        _token.transfer(owner(), returnedAmount);
        _finalized = true;
        emit Abort(_msgSender(), returnedAmount);
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual returns (bool) {
        return interfaceId == type(IKannaStockOption).interfaceId;
    }

    modifier initialized() {
        require(_initialized, "KannaStockOption: contract is not initialized");
        _;
    }
}

// File: contracts/KannaStockOptionManager.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.21;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import {KannaStockOption} from "./KannaStockOption.sol";
import {IKannaStockOption} from "./interfaces/IKannaStockOption.sol";

/**
 *   __
 *  |  | ___\|/_     ____    ____  _\|/_
 *  |  |/ /\__  \   /    \  /    \ \__  \
 *  |    <  / __ \_|   |  \|   |  \ / __ \_
 *  |__|_ \(____  /|___|  /|___|  /(____  /
 *       \/     \/      \/      \/      \/
 *            __                    __                       __   .__
 *    _______/  |_   ____    ____  |  | __   ____  ______  _/  |_ |__|  ____    ____
 *   /  ___/\   __\ /  _ \ _/ ___\ |  |/ /  /  _ \ \____ \ \   __\|  | /  _ \  /    \
 *   \___ \  |  |  (  <_> )\  \___ |    <  (  <_> )|  |_> > |  |  |  |(  <_> )|   |  \
 *  /____  > |__|   \____/  \___  >|__|_ \  \____/ |   __/  |__|  |__| \____/ |___|  /
 *       \/                     \/      \/         |__|                            \/
 *
 *  @title KNN Stock Option Manager
 *  @author KANNA Team
 *  @custom:github  https://github.com/kanna-coin
 *  @custom:site https://kannacoin.io
 *  @custom:discord https://discord.kannacoin.io
 */
contract KannaStockOptionManager is Ownable {
    uint16 private _nonce;
    mapping(address => bool) private _contractsMap;
    address[] private _contracts;

    address public contractTemplate;

    event ContractTemplateUpdated(
        address contractAddress
    );

    event ContractRegistered(
        address contractAddress,
        uint16 nonce
    );

    event ContractUnregistered(
        address contractAddress
    );

    /**
     * @dev Update StockOptionPool contract template
     */
    function updateTemplate(address _contractTemplate) public onlyOwner {
        require(
            IKannaStockOption(_contractTemplate).supportsInterface(type(IKannaStockOption).interfaceId),
            "`_contractTemplate` needs to implement `IKannaStockOption` interface"
        );

        contractTemplate = _contractTemplate;

        emit ContractTemplateUpdated(contractTemplate);
    }

    /**
     * @dev Deploy a new contract
     */
    function deployContract() public onlyOwner hasTemplateDefined {
        bytes32 _salt = keccak256(abi.encodePacked(_nonce, block.timestamp));

        address newContract = createClone(_salt, contractTemplate);

        _registerContract(newContract);
    }

    /**
     * @dev Register a new contract
     */
    function registerContract(address _contract) public onlyOwner {
        require(
            IKannaStockOption(_contract).supportsInterface(type(IKannaStockOption).interfaceId),
            "`_contract` needs to implement `IKannaStockOption` interface"
        );

        _registerContract(_contract);
    }

    /**
     * @dev Register a new contract
     */
    function registerContractUnsafe(address _contract) public onlyOwner {
        _registerContract(_contract);
    }

    /**
     * @dev Unregister a contract
     */
    function unregisterContract(address _contract) public onlyOwner {
        require(_contractsMap[_contract], "Contract not registered");

        _contractsMap[_contract] = false;

        for (uint256 i; i<_contracts.length; i++) {
            if (_contracts[i] == _contract) {
                _contracts[i] = _contracts[_contracts.length - 1];
                _contracts.pop();
                break;
            }
        }

        emit ContractUnregistered(_contract);
    }

    function contracts() public view returns (address[] memory) {
        return _contracts;
    }

    function hasContract(address _contract) public view returns (bool) {
        return _contractsMap[_contract];
    }

    function totalVested() public view returns (uint256) {
        uint256 _totalVested = 0;

        for (uint i=0; i<_contracts.length; i++) {
            try IKannaStockOption(_contracts[i]).totalVested() returns(uint256 contractTotalVested) {
                _totalVested += contractTotalVested;
            } catch Error(string memory /*reason*/) {
            } catch (bytes memory /*lowLevelData*/) {}
        }

        return _totalVested;
    }

    function vestingForecast(uint256 date) public view returns (uint256) {
        uint256 _vestingForecast = 0;

        for (uint i=0; i<_contracts.length; i++) {
            try IKannaStockOption(_contracts[i]).vestingForecast(date) returns(uint256 contractVestingForecast) {
                _vestingForecast += contractVestingForecast;
            } catch Error(string memory /*reason*/) {
            } catch (bytes memory /*lowLevelData*/) {}
        }

        return _vestingForecast;
    }

    function availableToWithdraw() public view returns (uint256) {
        uint256 _availableToWithdraw = 0;

        for (uint i=0; i<_contracts.length; i++) {
            try IKannaStockOption(_contracts[i]).availableToWithdraw() returns(uint256 contractAvailableToWithdraw) {
                _availableToWithdraw += contractAvailableToWithdraw;
            } catch Error(string memory /*reason*/) {
            } catch (bytes memory /*lowLevelData*/) {}
        }

        return _availableToWithdraw;
    }

    // creates clone using minimal proxy
    function createClone(bytes32 _salt, address _target) internal returns (address _result) {
        bytes20 _targetBytes = bytes20(_target);

        assembly {
            let clone := mload(0x40)
            mstore(clone, 0x3d602d80600a3d3981f3363d3d373d3d3d363d73000000000000000000000000)
            mstore(add(clone, 0x14), _targetBytes)
            mstore(add(clone, 0x28), 0x5af43d82803e903d91602b57fd5bf30000000000000000000000000000000000)
            _result := create2(0, clone, 0x37, _salt)
        }

        require(_result != address(0), "Create2: Failed on minimal deploy");
    }

    /**
     * @dev Register a new contract
     */
    function _registerContract(address _contract) private {
        require(_contractsMap[_contract] == false, "Contract already registered");

        _contracts.push(_contract);
        _contractsMap[_contract] = true;

        emit ContractRegistered(_contract, ++_nonce);
    }

    modifier hasTemplateDefined() {
        require(contractTemplate != address(0), "Contract template not defined");
        _;
    }
}